

1 Zadání příkladu

- Dva hmotné body se pohybují rovnoměrně přímočaře, první z bodu $A = [0, 1]$ m rychlostí $v_a = (3, -2) \text{ m} \cdot \text{s}^{-1}$, druhý z bodu $B = [0, -1]$ m rychlostí $v_b = (6, 4) \text{ m} \cdot \text{s}^{-1}$. Určete
- 1) průsečík trajektorií obou bodů,
 - 2) čas, kdy jsou si oba body nejbližší,
 - 3) nejmenší vzdálenost obou bodů

2 Postup řešení příkladu

- 1) Nejprve si parametricky vyjádříme oba pohyby ve tvaru $X = A + v \cdot t$. Průsečík trajektorií obou bodů určíme z podmínek, že x-ové a y-ové souřadnice v průsečíku jsou pro obě trajektorie shodné. Na parametrické vyjádření pohybu se budeme dívat jako na parametrické rovnice přímek. Proto parametr t v druhé rovnici si označíme s (jinak by rovnice nedávaly smysl). Když známe hodnotu t (a s), kdy se přímky protnou, zpětným dosazením do parametrického vyjádření získáme souřadnice průsečíku (pro kontrolu dosazení provedeme pro oba parametry). Pro další počítání parametr v druhé rovnici přepíšeme opět na t .
- 2) Pro určení času, kdy jsou si oba body nejbližší, si nejprve musíme určit závislost vzdálenosti obou bodů na čase. Tu určíme klasicky jako vzdálenost dvou bodů s tím rozdílem, že vzdálenost obou bodů závisí na čase (parametru t). Čas, kdy jsou si oba body nejbližší t_{\min} následně určíme z podmínky, že první derivace musí být rovna 0. To, že se jedná o minimum si ověříme graficky (šlo by taktéž použít znaménko druhé derivace).
- 3) Nejmenší vzdálenost určíme dosazením času t_{\min} do vztahu pro vzdálenost obou bodů.

Alternativní způsob řešení

Nyní tento příklad vyřešíme geometricky a graficky. Ze znalosti parametrického vyjádření z bodu 1) a závislosti vzdálenosti bodů na čase z bodu 2) nakreslíme příslušné grafy. Pro odečítání je lepší mít grafy co největší, abychom jsme se dopouštěli menší chyby měření.

- 1) Průsečík trajektorií obou bodů získáme odečtením příslušné hodnoty z grafu.
- 2), 3) Čas, kdy jsou si oba body nejbližší, získám pomocí grafu závislosti vzdálenosti obou bodů na čase. Také si můžeme rovnou nechat vypisovat hodnoty při daném parametru t .

3 Řešení příkladu

- Nastavení přesnosti vypisovaných číselných hodnot a zavolání balíků na kreslení grafů

```
(%i126) fpprintprec:4$
        load(draw)$
```

3.1 ad 1)

- Parametrické vyjádření pohybů (přímek)

```
(%i128) x1 : [0, 1] + [3, -2]*t;
        x2 : [0, -1] + [6, 4]*t;
        x1 : x1[1]$
        y1 : x1[2]$
        x2 : x2[1]$
        y2 : x2[2]$
(%o128) [3 t, 1 - 2 t]
(%o129) [6 t, 4 t - 1]
```

✓ Přepsání parametu v druhé rovnici

```
(%i134) x2 : subst(t = s, x2);
        y2 : subst(t = s, y2);
(%o134) 6 s
(%o135) 4 s - 1
```

✓ Určení parametrů t a s

```
(%i136) kill(t,s);
        parametry : solve([x1 = x2, y1 = y2], [t, s]);
        rhs(parametry[1][1]);
        rhs(parametry[1][2]);
(%o136) done
(%o137) [ [ t = 1/2, s = 1/4 ] ]
(%o138) 1/2
(%o139) 1/4
```

✓ Průsečík trajektorií obou bodů

```
(%i140) subst(t = rhs(parametry[1][1]), [x1, y1]);
        subst(s = rhs(parametry[1][2]), [x2, y2]);
(%o140) [ 3/2, 0 ]
(%o141) [ 3/2, 0 ]
```

✓ Přepsání parametu v druhé rovnici zpět

```
(%i142) x2 : subst(s = t, x2);
        y2 : subst(s = t, y2);
(%o142) 6 t
(%o143) 4 t - 1
```

□ 3.2 ad 2)

✓ Vzorec pro vzdálenost


```
(%i144) d:ratsimp(sqrt((x2-x1)^2+(y2-y1)^2));
(%o144)  $\sqrt{45 t^2 - 24 t + 4}$ 
```

Čas, kdy jsou body nejblíže

```
(%i145) tmin:rhs(solve(diff(d,t)=0,t)[1]);
          tmin,numer;
(%o145)  $\frac{4}{15}$ 
(%o146) 0.27
```

Grafické ověření, že se jedná o minimum

```
(%i147) wxplot2d(d,[t,0,1.5]);
```



3.3 ad 3)

Mimimální vzdálenost bodů

```
(%i148) dmin:subst(t=tmin,d);
          dmin,numer;
(%o148)  $\frac{2}{\sqrt{5}}$ 
(%o149) 0.89
```


3.4 Alternativní způsob řešení

Bohužel Maxima zatím neumožňuje interaktivně měnit parametry, následující grafy jsou proto udělány jako náhrada animací

```
(%i150) x1(t):=''x1$
        y1(t):=''y1$
        x2(t):=''x2$
        y2(t):=''y2$
        d(t):=''d$
```


```
(%i155) t1:0$
        load(draw)$
        wxdraw2d(pic_width = 650, pic_height = 400,
        key="Trajektorie bodu A",color=blue,parametric(x1(t),y1(t),t,0,3),
        key="Poloha bodu A v t=0 s",color=blue,point_size=1,point_type=7,
        points([[x1(t1),y1(t1)]]),
        key="Trajektorie bodu B",color=red,parametric(x2(t),y2(t),t,0,3),
        key="Poloha bodu B v t=0 s",color=red,point_size=1,point_type=7,
        points([[x2(t1),y2(t1)]]),
        xaxis=true,yaxis=true,xlabel="x",ylabel="y",
        xrange=[0,6], yrange=[-2,3],
        point_size=5,point_type=5,
        title= "Poloha v rovině")$
```

```
(%t157)
```




```
(%i158) wxdraw2d(pic_width = 650, pic_height = 400,  
key="Vzdálenost bodů",color=brown,explicit(d(t),t,0,1),  
key="Vzdálenost bodů v t=0 s",color=green,point_size=1,  
point_type=7, points([[t1,d(t1)]]),  
xrange=[0,1], yrange=[0,6],  
point_size=5,point_type=5,  
title= "Vzdálenost bodů")$
```

```
(%t158)
```



```
(%i159) wxdraw2d(pic_width = 650, pic_height = 400,
  key="Trajektorie bodu A",color=blue,parametric(x1(t),y1(t),t,0,3),
  key="Poloha bodu A po kroku 0.1 s",color=blue,point_size=1,
  point_type=7, points(makelist([x1(t1/10),y1(t1/10)],t1,0,20)),
  key="Trajektorie bodu B",color=red,parametric(x2(t),y2(t),t,0,3),
  key="Poloha bodu B po kroku 0.1 s",color=red,point_size=1,
  point_type=7, points(makelist([x2(t1/10),y2(t1/10)],t1,0,10)),
  xaxis=true,yaxis=true,xlabel="x",ylabel="y",
  xrange=[0,6], yrange=[-2,3],
  point_size=5,point_type=5,
  title= "Poloha v rovině")$

(%t159)
```



```
(%i160) wxdraw2d(pic_width = 650, pic_height = 400,  
key="Vzdálenost bodů",color=brown,explicit(d(t),t,0,1),  
key="Vzdálenost bodů po kroku 0.1 s",color=green,  
point_size=1,point_type=7,  
points(makelist([t1/10,d(t1/10)],t1,0,10)),  
xrange=[0,1], yrange=[0,6],  
point_size=5,point_type=5,  
title= "Vzdálenost bodů")$
```

```
(%t160)
```

